

# Eigenvalue problems

---

## Main idea and formulation in the linear algebra

The word "eigenvalue" stems from the German word "Eigenwert" that can be translated into English as "Its own value" or "Inherent value". This is a value of a parameter in the equation or system of equations for which this equation has a nontrivial (nonzero) solution. Mathematically, the simplest formulation of the eigenvalue problem is in the linear algebra. For a given square matrix  $A$  one has to find such values of  $\lambda$ , for which the equation (actually the system of linear equations)

$$A \cdot X = \lambda X \tag{1}$$

has a nontrivial solution for a vector (column)  $X$ . Moving the right part to the left, one obtains the equation

$$(A - \lambda I) \cdot X = 0,$$

where  $I$  is the identity matrix having all diagonal elements one and nondiagonal elements zero. This matrix equation has nontrivial solutions only if its determinant is zero,

$$\text{Det}[A - \lambda I] = 0.$$

This is equivalent to a  $N$ th order algebraic equation for  $\lambda$ , where  $N$  is the rank of the matrix  $A$ . Thus there are  $N$  different eigenvalues  $\lambda_n$  (that can be complex), for which one can find the corresponding eigenvectors  $X_n$ . Eigenvectors are defined up to an arbitrary numerical factor, so that usually they are normalized by requiring

$$X_n^{T*} \cdot X_n = 1,$$

where  $X^{T*}$  is the row transposed and complex conjugate to the column  $X$ .

It can be proven that eigenvectors that belong to different eigenvalues are orthogonal, so that, more generally than above, one has

$$X_m^{T*} \cdot X_n = \delta_{mn}.$$

Here  $\delta_{mn}$  is the Kronecker symbol,

$$\delta_{mn} = \begin{cases} 1, & m = n \\ 0, & m \neq n. \end{cases}$$

An important class of square matrices are Hermitean matrices that satisfy

$$A^{T*} = A.$$

Eigenvalues of Hermitean matrices are real. A real Hermitean matrix is just a symmetric matrix,  $A^T = A$ . For such matrices eigenvectors can be chosen real.

### ■ Matrix eigenvalue problem in Mathematica

*Mathematica* offers a solver for the matrix eigenvalue problem. If one is interested in eigenvalues only, one can use the command `Eigenvalues[...]`. Eigenvectors are computed by `Eigenvectors[...]`, while both eigenvalues and eigenvectors are computed by the command `EigenSystem[...]`. Let us illustrate how it works for a real symmetric matrix

$$A = \begin{pmatrix} a & b \\ b & -a \end{pmatrix};$$

Its eigenvalues are given by

**Eigenvalues[A]**

$$\{-\sqrt{a^2 + b^2}, \sqrt{a^2 + b^2}\}$$

Its eigenvectors are given by

**Eigenvectors[A]**

$$\left\{ \left\{ -\frac{-a + \sqrt{a^2 + b^2}}{b}, 1 \right\}, \left\{ -\frac{-a - \sqrt{a^2 + b^2}}{b}, 1 \right\} \right\}$$

that is,

**X<sub>i</sub> := Eigenvectors[A][[i]]**

These two eigenvectors are orthogonal to each other

**X<sub>1</sub>.X<sub>2</sub> // Simplify**

0

However, they are not normalized

**X<sub>1</sub>.X<sub>1</sub> // Expand // Factor**

$$\frac{2 \left( -a^2 - b^2 + a \sqrt{a^2 + b^2} \right)}{b^2}$$

To see which eigenvector corresponds to each eigenvalue, one has to use the command Eigensystem

**ESys = Eigensystem[A]**

$$\left\{ \left\{ -\sqrt{a^2 + b^2}, \sqrt{a^2 + b^2} \right\}, \left\{ \left\{ -\frac{-a + \sqrt{a^2 + b^2}}{b}, 1 \right\}, \left\{ -\frac{-a - \sqrt{a^2 + b^2}}{b}, 1 \right\} \right\} \right\}$$

The first part of this List are eigenvalues and the second part are eigenvectors. One can better see the correspondence in the form

**TableForm[Transpose[ESys]]**

$$\begin{array}{cc} -\sqrt{a^2 + b^2} & -\frac{-a + \sqrt{a^2 + b^2}}{b} \\ & 1 \\ \sqrt{a^2 + b^2} & -\frac{-a - \sqrt{a^2 + b^2}}{b} \\ & 1 \end{array}$$

*Mathematica* also solves matrix eigenvalue problems numerically, that is the only way to go for big matrices. For instance,**ESys = Eigensystem[A /. {a → 1., b → 2.}]**

```
{{-2.23607, 2.23607}, {{0.525731, -0.850651}, {-0.850651, -0.525731}}}
```

The numerical eigenvectors

**X<sub>i</sub> := ESys[[2]][[i]]**

are orthonormal

$\mathbf{x}_1 \cdot \mathbf{x}_1$   
 $\mathbf{x}_1 \cdot \mathbf{x}_2$   
 1.  
 0.

For a complex Hermitian matrix eigenvalues are indeed real, although eigenvectors are complex

```
TableForm[Transpose[Eigensystem[{{2, I}, {I, -2}}]]]

```

$$\begin{array}{cc}
 -\sqrt{3} & \mathbf{i}(-2 + \sqrt{3}) \\
 & 1 \\
 \sqrt{3} & -\mathbf{i}(2 + \sqrt{3}) \\
 & 1
 \end{array}$$

## Eigenvalue problem for systems of linear ODEs on time

The importance of the eigenvalue problem in physics (as well as in engineering and other areas) is that it arises on the way of solution of systems of linear ordinary differential equations with constant coefficients. We have already obtained the solution for the harmonic oscillator on this way in the chapter on differential equations.

Every linear ODE or a system of ODEs can be represented in the basic matrix form with a constant matrix  $A$

$$X'[t] + A \cdot X[t] = 0,$$

$X$  being a vector. (We drop the inhomogeneous term.) Searching for the solution in the form

$$X[t] = X_0 e^{-\lambda t},$$

one arrives at the eigenvalue problem Eq. (1) with  $X \Rightarrow X_0$ . After finding eigenvalues  $\lambda_n$  and normalized eigenvectors  $X_{0n}$  by linear algebra, one can write down the general solution of the equation as a linear superposition of all these solutions,

$$X[t] = \sum_{n=1}^N C_n X_{0n} e^{-\lambda_n t},$$

where  $C_n$  are integration constants that can be found from the initial conditions.

## Eigenvalue problems for PDE

In physical problems described by partial differential equations, eigenvalue problems usually arise due to boundary conditions.

### ■ Standing waves in a pipe

Consider, as an example, the wave equation for the pressure change (see Waves) in the 1d region  $0 \leq x \leq L$ ,

$$\partial_t^2 \delta P - c^2 \partial_x^2 \delta P = 0.$$

If we consider a pipe with both ends open to the atmosphere, the boundary conditions are

$$\delta P[0, t] = 0, \quad \delta P[L, t] = 0$$

because the pressure at the open end (practically) merges with the constant atmospheric pressure. Searching for  $\delta P$  in the form

$$\delta P[x, t] = \psi[x] \cos[\omega t + \phi_0]$$

one obtains the stationary wave equation

$$\partial_x^2 \psi + k^2 \psi = 0, \quad k = \omega / c,$$

$k$  being the wave vector. This is an eigenvalue problem because this equation has nontrivial solutions that satisfy the boundary conditions only for some values of  $k$  and thus of  $\omega$ . The general solution of the ODE above is

$$\psi[x] = C_1 \sin[kx] + C_2 \cos[kx].$$

Since  $\cos[kx]$  does not satisfy the BC at  $x = 0$ , the solution simplifies to

$$\psi[x] = C \sin[kx]$$

that describes a standing wave. Next, the BC at  $x = L$  requires  $\sin[kL] = 0$  from which one obtains the eigenvalues of the wave vector

$$k = k_n = \frac{\pi n}{L}, \quad n = 1, 2, 3, \dots$$

In terms of the wave length of the standing wave  $\lambda = 2\pi/k$  one has

$$\lambda_n = \frac{2L}{n}, \quad n = 1, 2, 3, \dots$$

Standing wave with  $n = 1$  is called fundamental wave, whereas those with  $n = 2, 3, \dots$  are called overtones or harmonics. For the frequencies of these waves  $f = \omega/(2\pi) = c/\lambda$  one has

$$f_n = \frac{cn}{2L}, \quad n = 1, 2, 3, \dots$$

The general solution of the wave equation for a pipe with both ends open is a linear superposition of all these solutions,

$$\delta P[x, t] = \sum_{n=1}^{\infty} C_n \sin[k_n x] \cos[\omega_n t + \phi_n], \quad k_n = \frac{\omega_n}{c} = \frac{\pi n}{L}, \quad n = 1, 2, 3, \dots \quad (2)$$

The coefficients  $C_n$  and phases  $\phi_n$  in this solution are arbitrary. Similar results can be obtained for a pipe with both ends closed (such as flute). If one end is closed and one is open (clarinet), the solution is somewhat different and only odd overtones exist. (Excercise). While the phases  $\phi_n$  are irrelevant for our ears (Ohm's psychoacoustical law), the amplitudes  $C_n$  define the quality of the sound via the relative weight of the overtones. This depends on how the music instrument is constructed and how it is played.

Eigenfunctions corresponding to different eigenvalues are orthogonal,

$$\int_0^L \sin[k_m x] \sin[k_n x] dx = \frac{L}{2} \delta_{mn}.$$

One can define the normalized eigenfunctions

$$\psi_n[x] = \sqrt{\frac{2}{L}} \sin[k_n x]$$


that satisfy

$$\int_0^L \psi_m[x] \psi_n[x] dx = \delta_{mn}.$$

Several lowest eigenfunctions are plotted below.

$$L = 1; \quad \psi_{n\mathbf{x}}[n, \mathbf{x}] = \sqrt{\frac{2}{L}} \operatorname{Sin}\left[\frac{\pi n}{L} \mathbf{x}\right];$$

Plot[Table[\(\psi\_{n\mathbf{x}}[n, \mathbf{x}]\), {n, 1, 4}], {x, 0, L}]



### ■ General formulation of the eigenvalue problem for PDE

In general, the eigenvalue problem for PDE can be formulated in the form

$$\hat{L} \psi[\mathbf{r}] = \lambda \psi[\mathbf{r}],$$

where  $\hat{L}$  is a differential operator. The best example is the stationary Schrödinger equation for a quantum particle

$$\hat{H} \psi = E \psi, \quad \hat{H} = -\frac{\hbar^2 \Delta}{2m} + U[\mathbf{r}], \quad (3)$$

where  $\hat{H}$  is the Hamilton operator. For a particle bound to a potential well, this equation has solutions satisfying boundary conditions only for select discrete values of the energy,  $E = E_n$ ,  $n = 0, 1, 2, \dots$ . These energy eigenvalues are labeled by the index  $n$  in the increasing order, and  $n = 0$  corresponds to the minimal energy, the so-called ground-state energy.

When a quantum particle undergoes a transition from the energy level (energy eigenstate)  $m$  to a lower energy level  $n$ , it emits a quantum of electromagnetic radiation (a photon), to conserve the total energy. The energy of the photon  $\varepsilon$  is related to its frequency  $\omega$  as  $\varepsilon = \hbar\omega$  (Max Planck), and from the energy conservation follows  $\hbar\omega = E_m - E_n$ . Spectroscopic investigations of the light emitted (and absorbed) by simple quantum systems such as hydrogen atoms have shown that only particular discrete frequencies obeying certain simple laws are present in the spectrum. This was at a contradiction with classical mechanics and electrodynamics that predicted a continuous energy spectrum. Physicists started to look for a mathematical tool that describes discreteness and made use of the eigenvalue problem. Werner Heisenberg proposed his matrix quantum mechanics operating with eigenvalues of matrices describing quantum systems. Erwin Schrödinger obtained his famous Schrödinger equation in which discrete energy levels of a quantum system arise as eigenvalues of a differential operator. The approaches of Heisenberg and Schrödinger are equivalent.

Solving the eigenvalue problem for a general differential operator, especially in  $2d$  and  $3d$ , is not an easy task, and there is no *Mathematica* solver for this problem at the moment. If the differential operator and boundary conditions possess special properties such as symmetry (e.g., spherical or cylindrical symmetry), the solution can be searched for in the form of a product of functions of different variables, so that the solution simplifies. This case is called "separation of variables".

Similarly to eigenvectors of matrices, eigenfunctions of differential operators can be complex. It can be shown that eigenfunctions corresponding to different eigenvalues are orthogonal. Also eigenfunctions corresponding to discrete eigenvalues can be normalized. Allowing for complex eigenfunctions, the orthonormality conditions takes the form

$$\int \psi_m^*[\mathbf{r}] \psi_n[\mathbf{r}] \, dV = \delta_{mn}, \quad (4)$$

where integration is carried out over the volume. An important mathematical theorem states that the whole set of eigenfunctions  $\psi_n[\mathbf{r}]$  of a differential operator satisfying boundary conditions forms a complete basis, so that any function  $f[\mathbf{r}]$  satisfying these BC can be expanded over the eigenfunctions as

$$f[\mathbf{r}] = \sum_{n=1}^{\infty} c_n \psi_n[\mathbf{r}]. \quad (5)$$

The expansion coefficients can be found using the orthonormality condition that yields

$$\int \psi_m^*[\mathbf{r}] f[\mathbf{r}] dV = \int \sum_{n=1}^{\infty} c_n \psi_m^*[\mathbf{r}] \psi_n[\mathbf{r}] dV = \sum_{n=1}^{\infty} c_n \delta_{mn} = c_m. \quad (6)$$

### ■ Quantum particle in a one-dimensional potential box

The stationary Schrödinger equation (6) for a quantum particle in a one-dimensional potential box  $0 \leq x \leq L$  has the form

$$\partial_x^2 \psi + k^2 \psi = 0, \quad k = \sqrt{2mE} / \hbar.$$

Everywhere in the box  $U = 0$  while outside the box  $U = \infty$ , so that the wave function outside the box is zero and by continuity it has to be zero at the box walls,

$$\psi[0, t] = 0, \quad \psi[L, t] = 0.$$

Mathematically this eigenvalue problem is equivalent to the problem of standing waves in a pipe considered above, except for another definition of the wave vector  $k$ . It has the same solution for the normalized eigenfunctions

$$\psi_n[x] = \sqrt{\frac{2}{L}} \sin[k_n x], \quad k_n = \frac{\pi n}{L}, \quad n = 1, 2, 3, \dots$$

The corresponding energy eigenvalues are given by

$$E_n = \frac{\hbar^2 k_n^2}{2m} = \frac{\pi^2 \hbar^2 n^2}{2mL^2}.$$

These energy levels are discrete, and a transition  $m \rightarrow n$  should be accompanied by emission of a photon of frequency

$$\omega_{mn} = \frac{E_m - E_n}{\hbar} = \frac{\pi^2 \hbar}{2mL^2} (m^2 - n^2).$$

Measuring all these frequencies, one can figure out the energy levels.

### ■ Quantum particle in a three-dimensional potential box and density of states

#### ■ Energy levels

Let us consider now a quantum particle in  $3d$  potential box  $0 \leq x \leq L_x, 0 \leq y \leq L_y, 0 \leq z \leq L_z$ . One can easily check that the solution with separated variables

$$\psi[x, y, z] \propto \sin[k_x x] \sin[k_y y] \sin[k_z z]$$

satisfies the stationary Schrödinger equation

$$\Delta \psi + k^2 \psi = 0$$

since  $k_x^2 + k_y^2 + k_z^2 = k^2$ . Also this solution satisfies the BC at  $x, y, z = 0$ , where  $\psi$  vanishes. Taking into account three BC at the other sides of the box finally yields

$$\psi_{n_x n_y n_z} [x, y, z] = \frac{2^{3/2}}{\sqrt{L_x L_y L_z}} \sin[k_x n_x x] \sin[k_y n_y y] \sin[k_z n_z z],$$

where the quantized wave vectors are given by

$$k_{\alpha n_\alpha} = \frac{\pi}{L_\alpha} n_\alpha, \quad n_\alpha = 1, 2, 3, \dots, \quad \alpha = x, y, z.$$

The discrete energy eigenvalues are parametrized by three integers,

$$E_{n_x n_y n_z} = \frac{\pi^2 \hbar^2}{2m} \left( \frac{n_x^2}{L_x^2} + \frac{n_y^2}{L_y^2} + \frac{n_z^2}{L_z^2} \right). \quad (7)$$

Another important case of separation of variables is the hydrogen atom which possesses a spherical symmetry. Rewriting Schrödinger equation in spherical coordinates allows to separate variables  $r$ ,  $\theta$ , and  $\phi$  and find analytical solutions for the energy levels and corresponding wave functions.

#### ■ Density of states

For potential boxes of sufficiently large sizes the quantum energy levels are very close to each other, so that one can consider the density of states  $\rho[E]$  as the number of energy levels  $dN_E$  in the energy interval  $dE$ ,

$$dN_E = \rho[E] dE.$$

For the 3d box the density of states is given by

$$\rho[E] = \frac{V}{(2\pi)^2} \left( \frac{2m}{\hbar^2} \right)^{3/2} \sqrt{E}, \quad V = L_x L_y L_z.$$

It is instructive to plot a histogram of the discrete energy levels given by Eq. (7) and to compare it with the formula above.

```

Lx = 91;      Ly = 100;      Lz = 110;
V = Lx Ly Lz;
nxMax = Lx;   nyMax = Ly;   nzMax = Lz;
nLevels = nxMax nyMax nzMax
ħ = 1;   M = 1.;


ρ[EE_] =  $\frac{V}{(2\pi)^2} \left(\frac{2M}{\hbar^2}\right)^{3/2} \sqrt{EE}$ ;

Enxnynz[nx_, ny_, nz_] =  $\frac{\pi^2 \hbar^2}{2M} \left(\frac{nx^2}{Lx^2} + \frac{ny^2}{Ly^2} + \frac{nz^2}{Lz^2}\right)$ ;

EMin = 0;
EMax =  $\frac{1}{3}$  Enxnynz[nxMax, nyMax, nzMax];
NE = 100; (* Number of energy bins dE *)
dE = (EMax - EMin) / NE; (* Width of energy bin dE *)
EList =
  Flatten[Table[Enxnynz[nx, ny, nz], {nx, 0, nxMax}, {ny, 0, nyMax}, {nz, 0, nzMax}]];
(* List of energy eigenvalues *)
EBinCounts =  $\frac{1}{dE}$  BinCounts[EList, {EMin, EMax, dE}];
(* Number of eigenvalues in energy bins dE divided by dE *)
DOSList = Table[{nE dE, EBinCounts[[nE]]}, {nE, 0, NE}];
(* Same with the energy values for the energy bins *)
Show[
  ListPlot[DOSList, PlotStyle -> {Orange, PointSize[0.02]}, AxesLabel -> {"E", "ρ[E]"},
  Plot[ρ[EE], {EE, EMin, EMax}, PlotStyle -> {Black, Thick}, AxesLabel -> {"E", "ρ[E]"}]
]

```

1 001 000



## ■ Quantum harmonic oscillator

Consider a one-dimensional quantum harmonic oscillator that is described by the Hamilton function

$$H = \frac{p^2}{2m} + U[x], \quad U[x] = \frac{m\omega_0^2}{2} x^2 \quad (8)$$

where  $\omega_0$  is the oscillator's frequency. The stationary Schrödinger equation  $E\psi = \hat{H}\psi$  for the oscillator reads

$$\left( \frac{\hbar^2}{2m} \frac{d^2}{dx^2} + E - U[x] \right) \psi = 0.$$

This is a second-order linear ODE with a variable coefficient. Boundary conditions require that  $\psi$  decreases to zero at  $x = \pm\infty$ . Solution of such differential equations is a more complicated task than for those with constant coefficients, and the analytical solution is usually expressed via special functions, if they are known. In this case an involved analytical solution yields the energy levels

$$E_n = \hbar\omega_0 (n + 1/2)$$

and normalized eigenfunctions

$$\psi_n[x] = \frac{1}{\sqrt{2^n n!}} \left(\frac{m\omega_0}{\pi\hbar}\right)^{1/4} \text{Exp}\left[-\frac{m\omega_0 x^2}{2\hbar}\right] H_n\left(\sqrt{\frac{m\omega_0}{\hbar}} x\right), \quad (9)$$

where  $H_n[x]$  are Hermite polynomials defined by

$$H_n[x] = (-1)^n e^{x^2} \frac{d^n}{dx^n} e^{-x^2}.$$

In *Mathematica* one can calculate Hermite polynomials straightforwardly and fast enough as

```
f[x_] = e^{-x^2};
MyHermiteH[n_, x_] := (-1)^n e^{x^2} Derivative[n][f][x] // Simplify

MyHermiteH[0, x]
MyHermiteH[1, x]
MyHermiteH[2, x]
MyHermiteH[3, x]

1
2 x
-2 + 4 x^2
4 x (-3 + 2 x^2)
```

etc., although there is a built-in function `HermiteH[n, x]`

```
HermiteH[0, x]
HermiteH[1, x]
HermiteH[2, x]
HermiteH[3, x]

1
2 x
-2 + 4 x^2
-12 x + 8 x^3
```

One can see that eigenfunctions of the quantum harmonic oscillator are, in fact, elementary functions. Still, Hermite and many other polynomials are put into the class of special functions by tradition. Let us plot the energy eigenvalues and eigenfunctions together with the potential.

```

Parameters = {m -> 1, h -> 1, w0 -> 1};
U[x_] = (m w0^2 x^2) / 2 /. Parameters
En[n_] = h w0 (n + 1 / 2) /. Parameters
ψnx[n_, x_] = (1 / Sqrt[2^n n!]) (m w0 / π h)^{1/4} Exp[-m w0 x^2 / 2 h] HermiteH[n, Sqrt[m w0 / h] x] /. Parameters


x^2
2

1
- + n
2

e^{-x^2/2} HermiteH[n, x]
π^{1/4} Sqrt[2^n n!]

xMax = 6; nMax = 16;
Show[
Plot[U[x], {x, -xMax, xMax}, PlotStyle -> {Black, Thick}], (* Potential energy *)
Plot[Table[En[n], {n, 0, nMax}], {x, -xMax, xMax}, PlotStyle -> Dashed],
(* Energy eigenvalues *)
Plot[Evaluate[Table[0.7 ψnx[n, x] + En[n], {n, 0, nMax}]], {x, -xMax, xMax},
PlotStyle -> Black] (* Eigenfunctions at the height of their eigenvalues *)
]

```



It is convenient to put eigenfunctions at the height of their eigenvalues for better association and separation. The numeric factor at eigenfunctions has been introduced to avoid overlap. Command Evaluate makes plotting faster by first evaluating eigenfunctions and then plotting them. Otherwise they are evaluated a every plot point.

One can see that even and odd eigenfunctions alternate. Similarly to plane quantum waves discussed in the chapter on Schrödinger equation, wave functions oscillate in the classically allowed region  $E - U[x] > 0$  and exponentially decrease in the classically prohibited region  $E - U[x] < 0$ .

As we know, a general solution of a second-order differential equation is a superposition of two independent functions. As in the case of a plane wave, in the classically prohibited region one of these functions is exponentially increasing while the other is exponentially decreasing. In this and similar problems, eigenfunctions decrease on both sides of the potential well. Other functions increase in the prohibited region and have to be rejected. It turns out that the solutions of the stationary Schrödinger equation that decrease on both sides exist only if  $E = E_n$ . For other energy values, both solutions of the DE increase at least on one side of the well and thus violate the boundary conditions.

This fact can be used to numerically find eigenvalues and eigenfunctions by the shooting method. One can start at some  $x_{\text{left}}$  left from the well setting any boundary conditions, say  $\psi[x_{\text{left}}] = 1$  and  $\psi'[x_{\text{left}}] = 0$  and solve the DE for any  $E$  to the right until some  $x_{\text{right}}$ . If  $\psi[x_{\text{right}}]$  is exponentially small,  $E$  is one of the energy eigenvalues, and if  $\psi[x_{\text{right}}]$  is exponentially large,  $E$  is not an eigenvalue.

```

Parameters = {m -> 1, ħ -> 1, ω₀ -> 1};
U[x_] =  $\frac{m \omega_0^2 x^2}{2}$  /. Parameters;
En[n_] = ħ ω₀ (n + 1 / 2) /. Parameters

xRight = 10; xLeft = -xRight;


IConds = {ψ[xLeft] == 1, ψ'[xLeft] == 0};

Sol[EE_] := NDSolve[
  Join[{ $\frac{\hbar^2}{2m} \psi''[x] + (EE - U[x]) \psi[x] == 0$  /. Parameters}, IConds], ψ, {x, xLeft, xRight}];
ψx[EE_?NumericQ, x_] := ψ[x] /. Sol[EE][[1]]
(* Without ?NumericQ FindMinimum below crashes *)

 $\frac{1}{2} + n$ 

```

Plotting shows that for our parameter choice the energy eigenvalues indeed are 0.5, 1.5, 2.5, etc., that is,  $E_n = n + 1/2$ .




Now one can plot eigenfunctions for the found eigenvalues

```

xMax = 5; xMin = -xMax;
Plot[Evaluate[ψx[0.5, x]], {x, xMin, xMax}, PlotRange → All]
Plot[Evaluate[ψx[1.5, x]], {x, xMin, xMax}, PlotRange → All]
Plot[Evaluate[ψx[2.5, x]], {x, xMin, xMax}, PlotRange → All]

```



These are essentially the same eigenfunctions as the analytical result considered above, only they are not normalized.

One could find the exact eigenvalues by minimizing  $\text{Log}[\text{Abs}[\psi_x[\text{EE}, \text{xRight}]]]$  starting from different values of EE.

```
FindMinimum[Log[Abs[ψx[EE, xRight]]], {EE, 0}]
```


FindMinimum::sdprec:

Line search unable to find a sufficient decrease in the function value with MachinePrecision digit precision. >>

```
{76.1993, {EE → 0.5}}
```

Finding exact eigenvalues by numerical minimization is important because this shooting method can be applied to potentials that do not allow for an analytical solution. For instance, if one replaces  $x^2 \Rightarrow x^4$  in  $U[x]$ , one obtains, for the same parameters,

```
Plot[Log[Abs[ψ[x][EE, xRight]]], {EE, 0, 5}]
```



These nonequidistant energy eigenvalues are not described by any analytical formula, so that finding them numerically is important. Note that power potentials  $U[x] \propto \left(\frac{x}{x_0}\right)^n$  approach the potential box  $-x_0 < x < x_0$  in the limit  $n \rightarrow \infty$ , where the analytical results have been found above.

### ■ Tunneling of a quantum particle

Let us now consider the potential  $U[x] = -ax^2 + bx^4$  with  $a, b > 0$  that has two symmetric minima at  $x = \pm x_0 = \pm \sqrt{a/(2b)}$ . It is convenient to represent this potential in the form

$$U[x] = \frac{m \omega_0^2}{8} \frac{(x^2 - x_0^2)^2}{x_0^2}. \quad (10)$$

Its second derivative at the minima is given by

$$U[x_{\pm}] = \frac{m \omega_0^2}{8} \frac{(x^2 - x_0^2)^2}{x_0^2};$$

$$\partial_{x,x} U[x] / \cdot x \rightarrow x_0$$

$$m \omega_0^2$$

that is the same as for a harmonic oscillator with frequency  $\omega_0$ . A classical particle of mass  $m$  would perform harmonic oscillations with this frequency near the minima of this potential.

Now the solution

```


Parameters = {m → 1, ħ → 1, ω0 → 1, x0 → 3.5};
U[x_] =  $\frac{m \omega_0^2}{8} \frac{(x^2 - x_0^2)^2}{x_0^2}$  /. Parameters;
xRight = 7; xLeft = -xRight;

IConds = {ψ[xLeft] == 1, ψ'[xLeft] == 0};

Sol[EE_] := NDSolve[
  Join[ $\left\{ \frac{\hbar^2}{2m} \psi''[x] + (EE - U[x]) \psi[x] == 0 \right.$  /. Parameters}, IConds], ψ, {x, xLeft, xRight}];
ψx[EE_?NumericQ, x_] := ψ[x] /. Sol[EE][[1]];
(* Without ?NumericQ FindMinimum below crashes *)


xMax = 6; xMin = -xMax; EMax = 2;
Plot[U[x], {x, xMin, xMax}];
Plot[Log[Abs[ψx[EE, xRight]]], {EE, 0, EMax}, PlotRange → All]

```



reveals that low-lying energy eigenvalues form doublets. To see the lowest doublet better one can zoom in:

```
Plot[Log[Abs[ψx[EE, xRight]]], {EE, 0.475, 0.479}, PlotRange → All]
```



One can find both eigenvalues

```
E0 = EE /. FindMinimum[Log[Abs[ψx[EE, xRight]]], {EE, 0}][[2]]
E1 = EE /. FindMinimum[Log[Abs[ψx[EE, xRight]]], {EE, 0.5}][[2]]
```

```
FindMinimum::sdprec:
```

```
Line search unable to find a sufficient decrease in the function value with MachinePrecision digit precision. >>
```

```
0.476188
```

```
FindMinimum::sdprec:
```

```
Line search unable to find a sufficient decrease in the function value with MachinePrecision digit precision. >>
```

```
0.478131
```

And their spitting

```
Δ = E1 - E0
```

```
0.00194363
```


To normalize the eigenfunctions, one at first has to calculate the norms of both eigenfunctions

$$N_0 = N \left[ \int_{x_{\text{Left}}}^{x_{\text{Right}}} \psi_x[E_0, x]^2 dx \right]; \quad N_1 = N \left[ \int_{x_{\text{Left}}}^{x_{\text{Right}}} \psi_x[E_1, x]^2 dx \right];$$

$$\psi_{x0}[x_] = \frac{1}{\sqrt{N_0}} \psi_x[E_0, x]; \quad \psi_{x1}[x_] = \frac{1}{\sqrt{N_1}} \psi_x[E_1, x];$$

Below the two normalized eigenfunctions are plotted together with the scaled potential energy and a energy levels.

```
xMax = 7; xMin = -xMax;
Show[
  Plot[Evaluate[{ψx0[x], ψx1[x]}],
    {x, xMin, xMax}, PlotRange → All, PlotStyle → {Blue, Red}],
  Plot[0.1 U[x], {x, -xMax, xMax}, PlotStyle → {Black, Thick}, PlotRange → {0, 1}],
  (* Potential energy *)
  Plot[0.1 E0, {x, -xMax, xMax}, PlotStyle → {Black, Dashed}],
  Plot[0.1 E1, {x, -xMax, xMax}, PlotStyle → {Black, Dashed}]
]
```



One can see that the eigenfunctions of the ground-state doublet are even and odd. According to a quantum-mechanical theorem, the eigenfunction of the ground state  $\psi_0[x]$  does not change its sign on  $x$ . They are localized in the wells with little probability to be under the barrier,  $x \sim 0$ . These two eigenfunctions can be interpreted as even and odd linear combinations of the states in the left and right wells that are hybridized via tunneling, an essentially quantum phenomenon. With increasing the width or/and the height of the barrier, the separation between energy eigenvalues in tunneling doublets (tunneling splitting) decreases very fast, and it becomes difficult to calculate the splitting numerically. Fortunately, in this case tunneling splitting can be calculated analytically with the help of the quasiclassical approximation.

### ■ Matrix quantum mechanics

Eigenfunctions  $\chi_n[\mathbf{r}]$  of a Hamiltonian satisfying boundary conditions form a complete basis that can be used to expand the solution of Schrödinger equation. The Hamiltonian to which the eigenfunctions belong can be the Hamiltonian of the problem or any other Hamiltonian. Let us call it  $\hat{H}_0$ . The eigenvalue expansion has the form

$$\Psi[\mathbf{r}, t] = \sum_{n=0}^{\infty} a_n[t] \chi_n[\mathbf{r}], \quad (11)$$

similarly to the pressure expansion for the standing waves in a pipe, Eq. (2). This expansion transforms the time-dependent Schrödinger equation to a system of ODEs for the coefficients  $a_n[t]$ . Substituting Eq. (11) into the Schrödinger equation, one obtains

$$\sum_{n=0}^{\infty} i\hbar \dot{a}_n[t] \chi_n[\mathbf{r}] = \sum_{n=0}^{\infty} a_n[t] \hat{H} \chi_n[\mathbf{r}], \quad (12)$$

Multiplicating this equation  $\chi_m^*[\mathbf{r}]$  and integrating it over the space taking into account orthonormality of the eigenfunctions, one obtains

$$i\hbar \dot{a}_m[t] = \sum_{n=0}^{\infty} H_{mn} a_n[t], \quad H_{mn} \equiv \int \chi_m^*[\mathbf{r}] \hat{H} \chi_n[\mathbf{r}] dV. \quad (13)$$

This is a system of a (generally) infinite number of linear homogeneous ODEs that are coupled to each other via the matrix elements  $H_{mn}$ . If  $\hat{H} = \hat{H}_0$ , one has  $H_{mn} = \epsilon_m \delta_{mn}$ , where  $\epsilon_m$  are eigenvalues corresponding to the eigenfunctions  $\chi_m$ . In this case equations decouple,  $i\hbar \dot{a}_m[t] = \epsilon_m a_m[t]$ , and the solution is  $a_m[t] = a_m[0] \text{Exp}\left[-\frac{i\epsilon_m}{\hbar} t\right]$ . However, this is a trivial case, and usually the Hamiltonian of the system  $\hat{H}$  is more complicated than  $\hat{H}_0$ , so that the matrix  $H_{mn}$  is non-diagonal.

Eq. (13) can be written in the matrix form

$$i\hbar \dot{A}[t] = H \cdot A[t], \quad (14)$$

where  $A[t]$  is the vector composed of  $a_m[t]$ . If matrix  $H$  is time-independent, its solution is the matrix exponential

$$A[t] = \text{Exp}\left[-\frac{iH}{\hbar} t\right] \cdot A[0]$$

describing time evolution of the initial state  $A[0]$ . This formula can be used to obtain the solution with the help of *Mathematica*. It should be noted that Heisenberg proposed his matrix quantum mechanics independently of Schrödinger equation.

Stationary states are described by the time dependence

$$A[t] = \text{Exp}\left[-\frac{iE}{\hbar} t\right] C.$$

Substituting this into Eq. (14), one arrives at the matrix eigenvalue problem

$$EC = H \cdot C \quad (15)$$

or, explicitly,

$$E c_m = \sum_{n=0}^{\infty} H_{mn} c_n \quad (16)$$

for the energy eigenvalues  $E$  that is equivalent to Eq. (1). If matrix  $H$  is infinite, there is an infinite number of eigenvalues. Practically, the functional basis used to build matrices is cut, so that only  $N$  lowest basis eigenfunctions of  $\hat{H}_0$  are used and the matrices are of rank  $N + 1$ . Some quantum-mechanical problems such that spin problems are described by finite matrices from the very beginning. An advantage of the matrix formalism is that it allows to obtain all eigenvalues and eigenfunctions as a result of a single computation.

#### ■ Particle in a double - well potential

Let us apply the matrix formalism to the problem of the quartic double-well potential, Eq. (10), and choose the basic formed by the eigenstates of the Hamiltonian  $\hat{H}_0$  of the harmonic oscillator, Eq. (8). The Hamiltonian of the problem can be written as

$$\hat{H} = \hat{H}_0 + \hat{V}, \quad \hat{V} = \frac{m \omega_0^2}{2} \left( \frac{(x^2 - x_0^2)^2}{4 x_0^2} - x^2 \right).$$

Such a separation of the Hamiltonian into two parts is typical, and in the case  $\hat{V}$  is small, there are perturbative analytical methods based on expansion in powers of  $\hat{V}$ . However, in our case  $\hat{V}$  is not small and the problem will be solved numerically. With the help of  $\hat{H}_{0mn} = \epsilon_m \delta_{mn}$  Eq. (16) becomes

$$(E - \epsilon_m) c_m = \sum_{n=0}^{\infty} V_{mn} c_n, \quad (17)$$

where  $\epsilon_m = \hbar \omega_0 (m + 1/2)$  are the energy eigenvalues of the harmonic oscillator and

$$V_{mn} \equiv \int_{-\infty}^{\infty} \chi_m[x] \hat{V} \chi_n[x] dx,$$

where  $\chi_n[x]$  are given by Eq. (9). There is an analytical expression for these matrix elements for general  $m$  and  $n$ , however, *Mathematica 7* cannot find it. Still, *Mathematica* can analytically calculate  $V_{mn}$  for any particular values of  $m$  and  $n$  that we need to build the Hamiltonian matrix. We will be labeling the eigenvalues of the problem with the index  $\nu$ , so that  $E_\nu$  are energy eigenvalues,  $c_{\nu n}$  are the basis expansion coefficients, and

$$\psi_\nu[x] \equiv \sum_{n=0}^{\infty} c_{\nu n} \chi_n[x]$$

are eigenfunctions of the problem. Below is *Mathematica* code that solves the matrix eigenvalue problem for a quartic potential using the harmonic oscillator basis.

```

(* Definitions *)
NBasis = 20; (* Number of the lowest oscillator eigenfunctions in the basis *)
M = 1;      (* Mass of the particle *)
ħ = 1;      (* Planck's constant *)
ω₀ = 1;     (* Frequency of oscillation near the minima of U[x] *)
x₀ = 3.5;   (* Positions of the minima of U[x] *)

U[x_] =  $\frac{M \omega_0^2}{8} \frac{(x^2 - x_0^2)^2}{x_0^2}$ ; (* Potential energy of the particle *)

U0[x_] =  $\frac{M \omega_0^2 x^2}{2}$ ; (* Potential energy of the harmonic oscillator *)
en[n_] = ħ ω₀ (n + 1/2); (* Energy levels of the harmonic oscillator *)

χnx[n_, x_] =  $\frac{1}{\sqrt{2^n n!}} \left(\frac{M \omega_0}{\pi \hbar}\right)^{1/4} \text{Exp}\left[-\frac{M \omega_0 x^2}{2 \hbar}\right] \text{HermiteH}\left[n, \sqrt{\frac{M \omega_0}{\hbar}} x\right]$ ;

(* Eigenfunctions of the harmonic oscillator *)
V[x] = U[x] - U0[x]; (* Perturbation Hamiltonian *)
Vmn[m_, n_] :=
  If[Abs[m - n] == 0 || Abs[m - n] == 2 || Abs[m - n] == 4,  $\int_{-\infty}^{\infty} \chi_{nx}[m, x] V[x] \chi_{nx}[n, x] dx, 0$ ];
(* Most of matrix elements of V[x] are zero, so do not calculate them *)
Hmn[m_, n_] := en[n] KroneckerDelta[m, n] + Vmn[m, n];
(* Elements of Hamiltonian matrix *)

(* Creating Hamiltonian matrix and solving matrix eigenvalue problem *)
Timing[ HMatrix = Table[Hmn[m, n], {m, 0, NBasis}, {n, 0, NBasis}]; ]
TableForm[Take[HMatrix, 9, 9]] (* Show part of the Hamiltonian matrix *)
Timing[ ES = Eigensystem[HMatrix]; ]
ES = Transpose[Sort[Transpose[ES]]]; (* Eigensystem sorted in ascending order *)
Eval = First[ES] (* Eigenvalues *)
EVecT = Last[ES]; (* This is a matrix of eigenvectors lying horizontally *)
EVec = Transpose[EVecT]; (* This is a matrix of eigenvectors standing vertically *)
TableForm[Chop[Take[EVecT.EVec, 9, 9]]] (* Check that eigenvectors are orthonormal *)

(* Finalizing the solution *)
Ev[v_] := Eval[[v + 1]]; (* Energy eigenvalues of the problem *)
Δ = Ev[1] - Ev[0]; (* Ground-state splitting *)
c[v_, n_] := EVecT[[v + 1, n + 1]] (* Expansion coefficients of the
eigenfunctions of the problem over the harmonic oscillator basis *)
ψvx[v_, x_] :=  $\sum_{n=0}^{NBasis} c[v, n] \chi_{nx}[n, x]$  (* Eigenfunctions of the problem *)

{23.922, Null}

1.6639    0        -0.508684  0        0.0124974  0        0        0        0
0         1.94452   0        -0.856072  0        0.027945  0        0        0
-0.508684  0        2.25574  0        -1.17532  0        0.0484022  0        0
0         -0.856072  0        2.59758  0        -1.4717  0        0.0739356  0
0.0124974  0        -1.17532  0        2.97003  0        -1.74656  0        0.104561
0         0.027945  0        -1.4717  0        3.37309  0        -2.00043  0
0         0         0.0484022  0        -1.74656  0        3.80676  0        -2.23354
0         0         0         0.0739356  0        -2.00043  0        4.27105  0
0         0         0         0         0.104561  0        -2.23354  0        4.76594

{0., Null}

```

```
{0.476188, 0.478132, 1.2695, 1.35141, 1.8394, 2.21869,
 2.71446, 3.255, 3.84311, 4.47189, 5.13829, 5.84678, 6.57393, 7.45094,
 8.21213, 9.51985, 10.3086, 12.3669, 13.2037, 16.5672, 17.4787}

1. 0 0 0 0 0 0 0 0
0 1. 0 0 0 0 0 0 0
0 0 1. 0 0 0 0 0 0
0 0 0 1. 0 0 0 0 0
0 0 0 0 1. 0 0 0 0
0 0 0 0 0 1. 0 0 0
0 0 0 0 0 0 1. 0 0
0 0 0 0 0 0 0 1. 0
0 0 0 0 0 0 0 0 1.
```


The splitting of the ground-state doublet

$$\Delta = E_{\nu}[1] - E_{\nu}[0]$$

0.00194448

is in an excellent accordance with the value obtained above by the shooting method. The eigenfunctions of the ground-state doublet, plotted together with the scaled potential energy and a group of low-lying energy levels

```
xMax = 7; xMin = -xMax;
Show[
Plot[Evaluate[{ψνx[0, x], ψνx[1, x]}],
{x, xMin, xMax}, PlotRange → All, PlotStyle → {Blue, Red}],
Plot[0.1 U[x], {x, -xMax, xMax}, PlotStyle → {Black, Thick}, PlotRange → {0, 1}],
(* Potential energy *)
Plot[Table[0.1 Eν[ν], {ν, 0, 12}], {x, -xMax, xMax}, PlotStyle → {Black, Dashed}]
]
```



are the same as above, except for the irrelevant signs. The splitting of the ground-state doublet is not seen in this plot, whereas the splitting of the second doublet is seen. The states above the barrier are not grouped into doublets, and their separation increases with energy.

## ■ Sparse matrices

We have seen above that the main time of the calculation goes to fill the Hamiltonian matrix with elements, whereas finding its eigensystem is very fast. Filling a matrix of rank  $N$  requires  $N^2$  operations, even if many its elements are zero. In our Hamiltonian matrix, only the elements with  $|m - n| = 0, 2, 4$  are nonzero. Such matrices in which most of the elements are zero are called sparse matrices. In *Mathematica* a special fast method of filling sparse matrices is employed. First, a matrix with all elements equal to zero is created by one step. Then nonzero elements are inserted at their places. Actually, it is only nonzero elements and the matrix dimensions that are stored in the memory. Thus, both memory usage and speed are dramatically improved. In the case of nearly-diagonal matrices, there are  $\sim N$  nonzero elements, so that the time to fill such a sparse matrix is proportional to  $N$  rather than to  $N^2$ .

*Mathematica* command to create a sparse matrix object is `SparseArray`. Let us create a  $5 \times 5$  matrix with diagonal elements equal to 1 and subdiagonal elements equal to  $\sqrt{2}$ .

```
MySparseMatrix =
  SparseArray[{{i_, j_} /; Abs[i - j] == 0 -> 1, {i_, j_} /; Abs[i - j] == 1 -> Sqrt[2]}, {5, 5}]
SparseArray[<13>, {5, 5}]
```

Here `/;` means "under the condition". The output object is named but its content is hidden because it is not a standard list. It can be visualized with the command `MatrixForm`:

```
MatrixForm[MySparseMatrix]
```

$$\begin{pmatrix} 1 & \sqrt{2} & 0 & 0 & 0 \\ \sqrt{2} & 1 & \sqrt{2} & 0 & 0 \\ 0 & \sqrt{2} & 1 & \sqrt{2} & 0 \\ 0 & 0 & \sqrt{2} & 1 & \sqrt{2} \\ 0 & 0 & 0 & \sqrt{2} & 1 \end{pmatrix}$$

One can apply all standard commands such as `Eigensystem` to sparse matrix objects. Let us now compare the time needed to create the Hamiltonian matrix above as a standard (dense) matrix and sparse matrix. (The preceding code must be run before so that *Mathematica* knows definitions.) For `NBasis = 100` one obtains

```
NBasis = 40;

Timing[ HMatrix = Table[Hmn[m, n], {m, 0, NBasis}, {n, 0, NBasis}]; ]
MatrixForm[Take[HMatrix, 5, 5]]

Timing[ HMatrix = SparseArray[
  {{m_, n_} /; Abs[m - n] == 0 || Abs[m - n] == 2 || Abs[m - n] == 4 -> Hmn[m - 1, n - 1]},
  {NBasis + 1, NBasis + 1}]; ]
MatrixForm[Take[HMatrix, 5, 5]]

{97.156, Null}


$$\begin{pmatrix} 1.6639 & 0 & -0.508684 & 0 & 0.0124974 \\ 0 & 1.94452 & 0 & -0.856072 & 0 \\ -0.508684 & 0 & 2.25574 & 0 & -1.17532 \\ 0 & -0.856072 & 0 & 2.59758 & 0 \\ 0.0124974 & 0 & -1.17532 & 0 & 2.97003 \end{pmatrix}$$


{118.86, Null}


$$\begin{pmatrix} 1.6639 & 0 & -0.508684 & 0 & 0.0124974 \\ 0 & 1.94452 & 0 & -0.856072 & 0 \\ -0.508684 & 0 & 2.25574 & 0 & -1.17532 \\ 0 & -0.856072 & 0 & 2.59758 & 0 \\ 0.0124974 & 0 & -1.17532 & 0 & 2.97003 \end{pmatrix}$$

```

For higher NBasis the gain in using sparse matrices will theoretically increase. This example is not the best, however, because the difference in speed between the two methods is marginal. The reason is that the standard method also uses the optimized definition of the matrix elements, in which their lengthy calculation via the integral is done only in the case they are nonzero, otherwise zeros are being put into the Hamiltonian matrix. A real gain in using sparse matrix objects can be achieved only for really big matrices in which calculation of nonzero elements is fast.

### ■ Quantum rotator and pendulum

Let us apply matrix quantum mechanics to the problem of a quantum pendulum that is described by the Hamiltonian

$$\hat{H} = \frac{\hbar^2 \hat{l}^2}{2 I} + U[\phi], \quad \hat{l} \equiv -i \frac{\partial}{\partial \phi}, \quad U[\phi] = Mga (1 - \text{Cos}[\phi]),$$

where  $\hbar \hat{l}$  is the quantum-mechanical angular momentum operator for a rotation around a fixed axis by the angle  $\phi$ ,  $I$  is the moment of inertia,  $M$  is the mass, and  $a$  is the distance between the center of mass and the pivot point. Classical pendulum performs oscillations for  $E < 2 Mga$  and rotations in positive or negative direction for  $E > 2 Mga$ . The frequency of small oscillations near the minimum of the potential energy is

$$\omega_0 = \sqrt{\frac{Mga}{I}}.$$

In the case  $U[\phi] = 0$  pendulum simplifies to rotator.

Since adding  $2\pi n$  to the angle does not change the physical position of the pendulum, the wave function satisfies the periodicity condition

$$\Psi[\phi, t] = \Psi[\phi + 2\pi n, t], \quad n = \pm 1, \pm 2, \dots$$

Thus we consider the wave functions in the interval  $-\pi \leq \phi \leq \pi$  and require  $\Psi[-\pi, t] = \Psi[\pi, t]$ . A suitable complete set of basis functions satisfying this boundary conditions consists of the eigenfunctions of the angular momentum

$$\hat{l} \chi_m = m \chi_m$$

that are given by

$$\chi_m = \frac{1}{\sqrt{2\pi}} e^{im\phi}, \quad m = 0, \pm 1, \pm 2, \dots$$

One can check that these functions are orthonormal. They are eigenfunctions of the stationary Schrödinger equation  $E\psi = \hat{H}\psi$  for the rotator with the rotational energies

$$E_m = \frac{\hbar^2 m^2}{2 I}.$$

For the pendulum, these functions are not eigenfunctions but can be used as a basis to formulate the matrix Schrödinger equation (16). The matrix elements of the Hamiltonian are given by

$$H_{mn} = \int_{-\pi}^{\pi} \chi_m^*[\phi] \hat{H} \chi_n[\phi] d\phi = \int_{-\pi}^{\pi} e^{-im\phi} \left( -\frac{\hbar^2}{2 I} \frac{\partial^2}{\partial \phi^2} + Mga(1 - \text{Cos}[\phi]) \right) e^{in\phi} d\phi = \left( \frac{\hbar^2 m^2}{2 I} + Mga \right) \delta_{mn} - \frac{Mga}{2\pi} \int_{-\pi}^{\pi} e^{-im\phi} \text{Cos}[\phi] e^{in\phi} d\phi.$$

Using  $\text{Cos}[\phi] = (e^{i\phi} + e^{-i\phi})/2$ , for the last term one obtains  $Mga(\delta_{m,n+1} + \delta_{m,n-1})/2$ , so that the matrix element becomes

$$H_{mn} = \left( \frac{\hbar^2 m^2}{2I} + Mga \right) \delta_{mn} - \frac{Mga}{2} (\delta_{m,n+1} + \delta_{m,n-1}).$$

Below is a *Mathematica* code for solving the eigenvalue problem for the quantum pendulum. Among the parameters, at least one number has to be real (such as  $M=1.$ ), so that *Mathematica* does not try to find eigenvalues analytically.

```
(* Definitions *)
NBasis = 300; (* Number of the lowest rotator eigenfunctions in the basis *)
M = 1.;      (* Mass of the pendulum *)
II = 300;    (* Moment of inertia of the pendulum *)
ħ = 1;      (* Planck's constant *)
g = 1;      (* Gravity acceleration *)
a = 1;      (* Lever arm of the gravity force *)

ω₀ = √(M g a / II); (* Frequency of small oscillations *)

χₘφ[m_, φ_] = 1 / √(2 π) e^(i m φ); (* Eigenfunctions of the rotator *)

en[m_] = (ħ² m² / (2 II) + M g a);

Vmn[m_, n_] = - (M g a / 2) (KroneckerDelta[m, n + 1] + KroneckerDelta[m, n - 1]);
Hmn[m_, n_] := en[m] KroneckerDelta[m, n] + Vmn[m, n];
(* Elements of Hamiltonian matrix *)

(* Creating Hamiltonian matrix and solving matrix eigenvalue problem *)
Timing[
  HMatrix = Table[Hmn[m, n], {m, -NBasis / 2, NBasis / 2}, {n, -NBasis / 2, NBasis / 2}];
  Print["Part of the Hamiltonian matrix:"]
  TableForm[Take[HMatrix, 9, 9]] (* Show part of the Hamiltonian matrix *)
  Timing[ES = Eigensystem[HMatrix];]
  ES = Transpose[Sort[Transpose[ES]]]; (* Eigensystem sorted in ascending order *)
  EVal = First[ES]; (* Eigenvalues *)
  EVecT = Last[ES]; (* This is a matrix of eigenvectors lying horizontally *)
  EVec = Transpose[EVecT]; (* This is a matrix of eigenvectors standing vertically *)
  TableForm[Chop[Take[EVecT.EVec, 9, 9]]];
  (* Check that eigenvectors are orthonormal *)

(* Finalizing the solution *)
Eμ[μ_] := EVal[[μ + 1]]; (* Energy eigenvalues of the problem *)
c[μ_, m_] := EVecT[[μ + 1, m + 1 + NBasis / 2]] (* Expansion coefficients
of the eigenfunctions of the problem over the rotator basis *)
ψμφ[μ_, φ_] := ∑_{m=-NBasis/2}^{NBasis/2} c[μ, m] χₘφ[m, φ] (* Eigenfunctions of the problem *)

Print["Energy levels:"]
LevelsToShow = 70;      LevelsToShow = Min[LevelsToShow, NBasis];
Take[EVal, LevelsToShow]
Show[
  ListPlot[Take[EVal, LevelsToShow], AxesLabel → {"μ", "Eμ"}],
  Plot[2 M g a, {x, 0, LevelsToShow}]
]
{0.765, Null}
```

Part of the Hamiltonian matrix:

```


38.5  -0.5  0      0      0      0      0      0      0
-0.5  38.0017 -0.5  0      0      0      0      0      0
0     -0.5  37.5067 -0.5  0      0      0      0      0
0     0     -0.5  37.015 -0.5  0      0      0      0
0     0     0     -0.5  36.5267 -0.5  0      0      0
0     0     0     0     -0.5  36.0417 -0.5  0      0
0     0     0     0     0     -0.5  35.56  -0.5  0
0     0     0     0     0     0     -0.5  35.0817 -0.5
0     0     0     0     0     0     0     -0.5  34.6067
    
```

{0.063, Null}

Energy levels:

```

{0.028763, 0.0860783, 0.14297, 0.199433, 0.255463, 0.311053, 0.3662, 0.420896,
0.475137, 0.528915, 0.582225, 0.635059, 0.68741, 0.739272, 0.790635, 0.841492,
0.891834, 0.941652, 0.990935, 1.03967, 1.08786, 1.13547, 1.1825, 1.22894, 1.27477,
1.31998, 1.36454, 1.40844, 1.45166, 1.49417, 1.53595, 1.57696, 1.61719, 1.65658,
1.6951, 1.7327, 1.76931, 1.80486, 1.83926, 1.8724, 1.90401, 1.93429, 1.96081,
1.99032, 2.0026, 2.04529, 2.04672, 2.10579, 2.10586, 2.17306, 2.17306, 2.24618,
2.24618, 2.32446, 2.32446, 2.40747, 2.40747, 2.49492, 2.49492, 2.5866, 2.5866,
2.68235, 2.68235, 2.78207, 2.78207, 2.88565, 2.88565, 2.99302, 2.99302, 3.10412}
    
```



One can see that the low-lying energy levels are equidistant, as the levels of the harmonic oscillator. Levels with  $E_\mu < 2$  Mga are oscillatory. Levels above this energy correspond to rotations over the barrier. These levels form doublets because of the two possible directions of the rotation. However, these doublets are split because of the essentially quantum-mechanical effect, the overbarrier reflection. For instance,

$$E_\mu[50] - E_\mu[49]$$

$$E_\mu[52] - E_\mu[51]$$

$$E_\mu[54] - E_\mu[53]$$

$$2.0047 \times 10^{-6}$$

$$4.45956 \times 10^{-8}$$


$$7.72255 \times 10^{-10}$$

Splitting of level doublets, as the overbarrier reflection, decreases very fast with the energy. At high energies one approximately has

$$E_\mu = \frac{\hbar^2(\mu/2)^2}{2I},$$

the energy levels of the rotator. The eigenfunctions of the problem are similar to those of the harmonic oscillator at low energies. However, at high energies the probability density  $|\psi_\mu|^2$  is oscillating because eigenfunctions describe superpositions of rotations in different directions.

```
Plot[Evaluate[{Abs[ψμφ[0, π x]]^2, Abs[ψμφ[20, π x]]^2, Abs[ψμφ[60, π x]]^2}],
{x, -1, 1}, PlotRange → All, PlotPoints → 30,
PlotStyle → {{Black, Thick}, {Black, Thin}, {Red, Thin}}, AxesLabel → {"φ/π", "|ψμ[φ]|^2"}]
```




To better see mixing of rotations in different directions, one can plot the coefficients  $c_{\mu m}$  for one the doublets

```

Eμ[54]
Eμ[53]
Eμ[54] - Eμ[53]
Plot[{c[53, Round[m]], c[54, Round[m]]}, {m, -LevelsToShow, LevelsToShow},
  PlotRange -> All, PlotStyle -> {{Blue, Dashed}, {Black}}, AxesLabel -> {"m", "cμm"}]
2.32446
2.32446
7.72255 × 10-10

```



One can see that for one of the eigenfunctions the coefficients  $c_{\mu m}$  are even in  $m$  (solid line) while for the other they are odd in  $m$  (dashed line), that is, both of them are mixtures of rotations in different directions.